Lab02: Analysis of finite difference schemes

1 Finite difference approximations

Mathematica implementation of finite difference calculus

```
In[1]:= x \[Element] Reals; $Assumptions = h > 0;
        Dp[f_[x_]] := f[x + h] - f[x];
        Dm[f_[x_]] := f[x] - f[x - h];
        d[f_[x_]] := f[x+h/2]-f[x-h/2];
        Dp[a_ f_[x_]]=a Dp[f[x]]; Dp[a_+b_]=Dp[a]+Dp[b]; Dp[a_ (b_+c_)]=a Dp[b]+a Dp[c]; Dp[a_]=0;
        Dm[a_ f_[x_]]=a Dm[f[x]]; Dm[a_+b_]=Dm[a]+Dm[b]; Dm[a_ (b_+c_)]=a Dm[b]+a Dm[c]; Dm[a_]=0;
         d[a_ f_[x_]]=a  d[f[x]];  d[a_+b_]= d[a]+ d[b];  d[a_ (b_+c_)]=a  d[b]+a  d[c];  d[a_]=0;
        fdSort[expr_]:=Assuming[{h > 0, x > 0}, Sort[Level[expr, 1], Level[#1, {Depth[#1] - 1}] < Level[#2,
        {Depth[#2] - 1}] &]];
        Q[x_]=Subscript[Q,x];
        Dp[f_[x_], p_] :=
         Together[Normal[1/h Series[Log[1 + z], {z, 0, p}]] /.
           Table[z^j -> Nest[Dp, f[x], j], {j, 1, p}]];
        Dm[f_[x_], p_] :=
         Together[Normal[-1/h Series[Log[1 - z], {z, 0, p}]] /.
           Table[z^j -> Nest[Dm, f[x], j], {j, 1, p}]];
        d[f_[x_], p_] := Together[Normal[2/h Series[ArcSinh[z/2], {z, 0, 2 p - 1}]] /.
          Table[z^j -> Nest[d, f[x], j], {j, 1, 2 p - 1}]];
        Dp[f_[x_], p_, k_] :=
         Together[Expand[(Normal[1/h Series[Log[1 + z], {z, 0, p}]])^k] /.
           Table[z^j -> Nest[Dp, f[x], j], {j, 1, p k}]];
        Dm[f_[x_], p_, k_] :=
         Together[Expand[(Normal[-1/h Series[Log[1 - z], {z, 0, p}]])^k] /.
           Table[z^j -> Nest[Dm, f[x], j], {j, 1, p k}]];
        d[f_[x_], p_, k_] := Together[Expand[(Normal[2/h Series[ArcSinh[z/2], {z, 0, 2 p - 1}]])^k]
        /. Table[z^j -> Nest[d, f[x], j], {j, 1, (2 p - 1)k}]];
        {qt1,qxx2,qxx4}={Dp[f[x],1],d[f[x],1,2],d[f[x],2,2]}
```

$$\left\{\frac{f(h+x)-f(x)}{h}, \frac{f(x-h)+f(h+x)-2f(x)}{h^2}, \frac{f(x-3h)-54f(x-2h)+783f(x-h)+783f(h+x)-54f(2h+x)+f(3h+x)-1460f(x)}{576h^2}\right\}$$

```
In[2]:= Series[qxx4,{h,0,4}]
```

$$f''(x)-\frac{3}{320}h^4 f^{(6)}(x)+O(h^5)$$

```
In[3]:=
```

Note that centered formulas for approximating the second derivative of comparable accuracy may be obtained by using a smaller stencil.

```
In[3]:= dc[f_[x_], p_, k_] := Module[{z,n,zn,ser,serc,fd},
          ser = Expand[(Normal[2/h Series[ArcSinh[z/2], {z, 0, 2 p - 1}]])^k];
          n = Exponent[ser,z]; zn = Coefficient[ser,z,n] z^n;
          serc = ser - zn;
          fd = serc /. Table[z^j -> Nest[d, f[x], j], {j, 1, (2 p - 1)k}];
          Return[Together[fd]]];
        qxx4=dc[f[x],2,2]
```

$$\frac{-f(x-2h)+16f(x-h)+16f(h+x)-f(2h+x)-30f(x)}{12h^2}$$

```
In[4]:= Series[qxx4,{h,0,4}]
```

$$f''(x)-\frac{1}{90}h^4 f^{(6)}(x)+O(h^5)$$

```
In[5]:=
```

2 Eigenvalues of matrix from semi-discretization

From the PDE $q_t = q_{xx}$ an ODE system

$$\mathbf{Q}'(t) = \frac{1}{h^2}\mathbf{A}\,\mathbf{Q}(t) \tag{1}$$

is obtained. The eigenvalues of $\mathbf{A}$ are necessary for stability analysis of various time integration schemes to solve (1). Since $\frac{1}{h^2}\mathbf{A} \cong \partial_x^2$, the eigenvectors of $\mathbf{A}$ are discrete approximations of the eigenfunctions of ∂_x^2.

2.1 Homogeneous Dirichlet boundary conditions

Separation of variables applied to $q_t = q_{xx}$, $q(t,x) = T(t)X(x)$ gives

$$\frac{T'}{T} = \frac{X''}{X} = -\lambda^2, \lambda \in \mathbb{R}, X(0) = X(1) = 0$$

with eigenfunctions $X_k(x) = \sin(k\pi x)$, $\lambda_k = k\pi$.

```
In[31]:= op = -X''[x];
         lftBC = DirichletCondition[X[x]==0,x==0];
         rgtBC = DirichletCondition[X[x]==0,x==1];
         DEigensystem[{op,lftBC,rgtBC},X[x],{x,0,1},7]
```

$$\begin{pmatrix} \pi^2 & 4\pi^2 & 9\pi^2 & 16\pi^2 & 25\pi^2 & 36\pi^2 & 49\pi^2 \\ \sin(\langle \text{pix} \rangle) & \sin(2\langle \text{pix} \rangle) & \sin(3\langle \text{pix} \rangle) & \sin(4\langle \text{pix} \rangle) & \sin(5\pi x) & \sin(6\pi x) & \sin(7\pi x) \end{pmatrix}$$

```
In[48]:= qxx2
```

$$\frac{f(x-h) + f(h+x) - 2f(x)}{h^2}$$

```
In[49]:= xk[x_] = Sin[k Pi x];
         Axki = qxx2 /. {f->xk, x-> i h}
```

$$\frac{-2\sin(\pi h i k) + \sin(\pi k(h i - h)) + \sin(\pi k(h i + h))}{h^2}$$

```
In[55]:= lambda2[h_]=FullSimplify[Axki/xk[i h]]
```

$$\frac{2(\cos(\pi h k) - 1)}{h^2}$$

```
In[56]:= Limit[lambda2[h],h->0]
```

$$\pi^2(-k^2)$$

```
In[57]:=
```

2.2 Semi-homogeneous Dirichlet boundary conditions

Separation of variables applied to $q_t = q_{xx}$, $q(t,x) = T(t)X(x)$ gives

$$\frac{T'}{T} = \frac{X''}{X} = -\lambda^2, \lambda \in \mathbb{R}, X(0) = 0, X(1) = 1$$

with eigenfunctions $X_k(x) = \sin[(4k+1)\pi x/2]$, $\lambda_k = (4k+1)\pi/2$.

```
In[57]:= qxx2
```

$$\frac{f(x-h) + f(h+x) - 2f(x)}{h^2}$$

```
In[58]:= xk[x_] = Sin[(4k+1) Pi x/2];
         Axki = qxx2 /. {f->xk, x-> i h}
```

$$\frac{-2\sin\left(\frac{1}{2}\pi h i (4k+1)\right) + \sin\left(\frac{1}{2}\pi(4k+1)(h i - h)\right) + \sin\left(\frac{1}{2}\pi(4k+1)(h i + h)\right)}{h^2}$$

```
In[59]:= lambda2[h_]=FullSimplify[Axki/xk[i h]]
```

$$-\frac{4\sin^2\left(\pi h k + \frac{\pi h}{4}\right)}{h^2}$$

```
In[60]:= Limit[lambda2[h],h->0]
```

$$-\frac{1}{4}(4\pi k + \pi)^2$$

```
In[61]:=
```